Tajny Skrypt Lab SML^1 v 0.33

Marcin Benke^2

benke@mimuw.edu.pl

redakcja i rozdział 7

Grzegorz Grudziński^2

gsg@mimuw.edu.pl

rozdział 8

Mikołaj Konarski^2

mikon@mimuw.edu.pl

pozostałe rozdziały

10 października 2000

1Nazwa jest historyczna.

2Instytut Informatyki, Uniwersytet Warszawski.

Spis rzeczy

1 Wyrażenia 3 2 Listy & stuff 5 3 Rekursja ogonowa, akumulatory i różne takie 8 4 Definiowanie typów 11 5 Funkcje wyższych rzedów 14 6 Funkcje jeszcze wyższych rzedów 16 7 Operacje na tekstach 19 8 Leniwość 21 9 Moduły: struktury i sygnatury 22 10 Moduły: funktory 26 A Jak zaliczyć laboratorium 28 B Rozwiazania niektórych zadań 30 C Jak pracować z kompilatorem SML-a 36

Ostrzeżenie

#include <std/disclaimer.h> ;-)

Ten dokument, zgodnie ze swym tytułem, nie tylko nie jest oficjalnym skryptem, ale co wiecej — jest tajny.

Ponadto nie sa to notatki do wykładu, tylko do pracowni. Wiedza podawana jest tu wyrywkowo i skrótowo, a nacisk położony jast na zadania.

Tajne Kolegium Redakcyjne

Rozdział 1

Wyrażenia

Program w jezyku funkcyjnym to wyrażenie, a wykonanie programu to obliczenie wartości wyrażenia. Przykłady wyrażeń:

3; (* Comment: semicolon ";" tells the interpreter *)
(1 + 2) * 1; (* to evaluate expressions. *)
("Hi! I’m the first element of this tuple", 3, false);
if ((2.3 > 5.4) orelse true) then "yes" else "no";
(fn n => 2 * n) 3;
fn x => x;

Wielkie i skomplikowane wyrażenia zapisuje sie wygodniej używajac konstrukcji let...in...end oraz deklaracji:

let
   val text = "I’m a string"
in
   text^". Yes, "^text^". Sure, "^text^"!"
end;
val twice = (fn n => 2 * n);
twice;
twice (twice 3);
fun twice’ n = 2 * n;
twice 3 = twice’ 3 andalso twice’ 4 = twice 4;

Przykład z pi pokazuje pewna istotna ceche deklaracji. Jaka jest wartość wyrażenia area 5.0 z ostatniej linii tego przykładu?

val pi = 3.14;
fun area r = pi * r * r;
area 5.0;
val pi = 1.0;
area 5.0;

Poteżnym narzedziem programowania funkcjnego jest rekurencja:

fun fact n = if n <= 1 then 1 else n * (fact (n - 1));
fun fib n = if n <= 1 then 1 else (fib (n - 1)) + (fib (n - 2));
fact (fib 4);
fib;

Zadanie

Zadanie 1.1 Oto definicja funkcji fib’:

fun fib’ n =
    let
        fun fib_pair n =
            if n <= 1 then (1, 1)
                      else let
                               val (j, k) = fib_pair (n - 1)
                           in
                               (j + k, j)
                           end
    in
        #1 (fib_pair n)
    end

(#1 oznacza wziecie pierwszego elementu tupla). Udowodnij przy pomocy indukcji, że dla każdego n

fib n = fib’ n.

Koszt czasowy fib n jest wykładniczy wzgledem n. Jaki jest koszt fib’ n?

Rozdział 2

Listy & stuff

Poznaliśmy tuple:

(2, "errr", 3.14);

Listy to coś zupełnie innego:

[2, 55, 15346];
["This", "is", "a", "list", ", yeah."];
[];

Te nawiasy kwadratowe to skrótowy sposób zapisu list. Tak naprawde listy konstruuje sie w ten sposób:

2::(55::(15346::nil));

lub bez niepotrzebnych nawiasów:

2::55::15346::nil;
"This"::"is"::"a"::"list"::", yeah."::nil;
nil;

Mówiac ściśle: infiksowy konstruktor :: dostaje jako argumenty wartość x i liste l, której elementy sa tego samego typu co x, i daje w wyniku l z doklejonym na poczatku elementem x. Na przykład (używajac mieszanej notacji)

1::[2, 3, 4] reprezentuje te sama liste co [1, 2, 3, 4]

Kilka przykładów:

val mylist = [1, 2, 3];
fun null l = (l = []);
null mylist;
null [];
fun interval (n, m) = if n > m then [] else n::interval (n+1, m);
interval (~31, 72);

Jest troche wbudowanych funkcji na listach, w szczególności hd oddajaca pierwszy element listy i tl, oddajaca liste bez pierwszego elementu.

val head = hd mylist
val tail = tl mylist
val what = head :: tail

Funkcje na listach można definiowac za pomoca pattern matchingu

fun null [] = true
  | null (head::tail) = false

a tu bedzie potrzebny wyjatek (exception):

exception Empty
fun hd [] = raise Empty
  | hd (head::tail) = head

Dalej już łatwo :-)

Zadania

Zadanie 2.1 Przy pomocy pattern matchingu napisz funkcje tl.

Zadanie 2.2 Używajac funkcji null, hd, tl, napisz funkcje sum, która dodaje do kupy wszystkie elementy danej listy liczb całkowitych. Zrób to samo przy pomocy pattern matchingu.

Zadanie 2.3 Używajac funkcji null, hd, tl, napisz funkcje length, która oblicza długość listy. Zrób to samo przy pomocy pattern matchingu.

Zadanie 2.4 Używajac funkcji null, hd, tl, napisz funkcje last, która daje w wyniku ostatni element listy. Zrób to samo przy pomocy pattern matchingu.

Zadanie 2.5 Napisz funkcje append, która łaczy dwie listy. Na przykład:

append ([1, 2], [3, 4, 6]) = [1, 2, 3, 4, 6]

Zadanie 2.6 Napisz funkcje take i drop, z parametrem (l,n), które odpowiednio biora lub odrzucaja pierwsze n elementów listy l. Upewnij sie, że dla każdego (l, n)

append (take (l, n), drop (l, n)) = l

Zadanie 2.7 Napisz, korzystajac z funkcji drop i hd, funkcje nth, która daje nty element listy. Czy funkcja działałaby szybciej, gdyby skorzystać z take i last?

Zadanie 2.8 Napisz funkcje flatten, która dostaje liste list, a daje w wyniku liste wszystkich elementów tych list (nie martw sie powtórzeniami czy kolejnościa).

Zadanie 2.9 Napisz quicksort.

Zadanie 2.10 Przy pomocy list napisz funkcje realizujace arytmetyke na DU- ŻYCH liczbach całkowitych. Użyj ich do policzenia 500! albo lepiej 1000!. Jaka jest złożoność liczenia n! w ten sposób?

Zadanie 2.11 Napisz to samo w C lub Pascalu. Porównaj czas zużyty na pisanie i wykonanie programu.

Poza tym można zrobić mergesort albo funkcje obliczajace wszystkie permutacje, kombinacje, etc. danej listy albo pare funkcji: zip i split; zip dostaje dwie listy liczb całkowitych a oddaje liste par: para pierwszych elementów obu list, potem drugich, itd, split jest funkcja (prawie) odwrotna.

Wszystkie napisane programy warto przetestować — jest to łatwe i przyjemne, bo używamy interpretera.

Zachecamy do wszelakich eksperymentów!

Rozdział 3

Rekursja ogonowa, akumulatory i różne takie

Czesto funkcje można zdefiniować na kilka istotnie rożnych sposobów. Sztandarowym przykładem jest funkcja dajaca n–ta liczbe Fibonacciego, wśród której najprostszych definicji istnieja drastyczne rożnice złożoności czasowej. Duża rodzine innych przykładów stanowia funkcje przy których definiowaniu można ale nie trzeba używać akumulatora tzn. dodatkowego parametru który kumuluje składowe wyniku.

Oto definicja silni bez akumulatora:

fun fact n = if n <= 1 then 1 else n * fact (n - 1)

i z akumulatorem:

fun fact_tr n =
    let
        fun fact_acc (acc, k) = if k <= 0 then acc
                                else fact_acc (acc * k, k - 1)
    in
        fact_acc (1, n)
    end

przy czym widać, że dla każdego n mamy

fact n = fact tr n = n!.

Jak wyglada obliczanie fact 4? Funkcja fact 4 wywołuje fact 3, zapamietujac, że wynik tego wywołania powinna pomnożyć przez 4, nim odda go jako swój wynik. Funkcja fact 3 wywołuje fact 2, zapamietujac, że wynik tego wywołania powinna pomnożyć przez 3. Funkcja fact 2 wywołuje fact 1, zapamietujac, że wynik tego wywołania powinna pomnożyć przez 2. Teraz fact 1 oddaje 1. Nastepnie fact 2 mnoży to przez 2 i oddaje rezultat tego mnożenia, czyli 2. Potem fact 3 mnoży to przez 3 i daje 6. Potem fact 4 mnoży to przez 4 i oddaje 24. A wiec kolejność mnożeń była nastepujaca: ((1 ∗ 2) ∗ 3) ∗ 4.

Twierdze, że kolejność mnożeń przy obliczaniu fact_tr 4 jest odwrotna, czyli ((4 ∗ 3) ∗ 2) ∗ 1. Zobaczmy: fact_tr 4 wywołuje fact_acc (1, 4), który po wykonaniu jednego mnożenia na rozruch wywołuje fact_acc (4, 3). Nastepnie fact_acc (4, 3) mnoży 4 przez 3 i wywołuje fact_acc (12, 2). Nastepnie fact_acc (12, 2) mnoży 12 przez 2 i wywołuje fact_acc (24, 1). Nastepnie fact_acc (24, 1) mnoży 24 przez 1 i wywołuje fact_acc (24, 0).

(*) A teraz fact_acc (24, 0) oddaje 24 i wynik ten zwracany jest już bez zmian przez kolejno fact_acc (24, 1), fact_acc (12, 2), fact_acc (4, 3),

fact_acc (1, 4) i fact_tr 4.

Skoro kolejność mnożeń jest odwrotna, fact i fact_tr rzeczywiście istotnie sie od siebie różnia. Różnic jest wiecej. Na przykład definicja fact jest niewatpliwie bardziej czytelna od fact_tr. A z drugiej strony w fact_tr wystepuje wyłacznie tzw. rekursja ogonowa (tail recursion), przez co fact_tr wykonuje sie nieco szybciej i zużywa nieco mniej pamieci niż fact, w którym rekursja nie jest ogonowa.

Kiedy funkcja zawiera rekursje ogonowa? Wtedy, kiedy w definicji funkcji wołanie rekurencyjne nie jest podwyrażeniem żadnego innego wyrażenia (z wyjatkiem drugiej lub trzeciej gałezi wyrażenia warunkowego). Jeśli rekursja jest ogonowa, wynik wykonania każdej funkcji wołanej jest przekazywany bez modyfikacji jako wynik funkcji wołajacej, tak jak w miejscu oznaczonym (*).

Każdy przywoity kompilator zastepuje sekwencje powrotów taka jak (*) jedna instrukcja skoku, a także używa jednego rekordu aktywacji dla wszystkich rekurencyjnych wcieleń funkcji. W ten sposób rekurencja jest automatycznie zastepowana iteracja.

Zadania

Zadanie 3.1 Napisz wersje funkcji length z rekursja ogonowa.

Zadanie 3.2 Napisz funkcje decimal, która dostaje liste cyfr i odaje liczbe jaka powstaje z tych cyfr, jeśli je ustawić za soba (or something). np.

decimal [1, 9, 9, 5] ma dać 1995.

Zadanie 3.3 Napisz dwie wersje funkcji reverse odwracajacej liste. Jedna z tych wersji ma działać w czasie liniowym.

Zadanie 3.4 Czy można napisać ogonowa wersje funkcji take? Jeśli tak, to czy warto jej używać? Odpowiedź uzasadnij pełnymi zdaniami.

Zadanie 3.5 Napisz wersje funkcji interval (z poprzedniego rozdziału) z rekursja ogonowa. Nie używaj reverse ;-).

Zadanie 3.6 Napisz funkcje real_max, która dostaje liste liczb rzeczywistych i oddaje najwieksza z nich. Napisz dwie istotnie różne wersje. Która wersja bardziej Ci sie podoba ?

Zadanie 3.7 To inna wersja funkcji quicksort:

fun quicksort [] = [] : int list
  | quicksort [x] = [x]
  | quicksort (a::rest) =
    let fun split(left,right,[]) =
            quicksort left @ (a::quicksort right)
          | split(left,right,x::l) =
            if x < a then split(x::left,right,l)
            else split(left,x::right,l)
    in split([],[],rest)
    end

Porównaj ja z wersja podana w rozwiazaniu zadania z pracowni 2. Czytelność, ogonowość, akumulatorowość, etc.

Jak zwykle eksperymenty surowo wskazane!

Rozdział 4

Definiowanie typów

Można definiować typy:

datatype friend = PAT | SUSIE | JOE
datatype gift = BOTTLES_OF_BEER of int | BOOK

a przy pomocy pattern matchingu na tych typach tworzyć funkcje:

fun give PAT = BOTTLES_OF_BEER(10)
  | give SUSIE = BOOK
  | give JOE = BOTTLES_OF_BEER(3)
fun like (BOTTLES_OF_BEER(n)) = n > 7
  | like BOOK = true

Typy moga być rekurencyjne:

datatype int_list = INT_NIL | INT_CONS of int * int_list
fun sum INT_NIL = 0
  | sum (INT_CONS(i, rest)) = i + sum rest

wzajemnie rekurencyjne:

datatype friend_rec = PAT | SUSIE | ANYONE_WHO_GIVES_ME of gift_rec
and gift_rec = BOTTLES_OF_BEER of int | PHOTO of friend_rec
fun love PAT = false
  | love SUSIE = true
  | love (ANYONE_WHO_GIVES_ME(a_gift)) = like a_gift
and like (BOTTLES_OF_BEER(n)) = n > 7
  | like (PHOTO(my_friend)) = love my_friend

polimorficzne :

datatype ’a option = SOME of ’a | NONE
val int_some_list = [SOME(5), NONE, SOME(10), SOME(3), NONE]
val bool_some_list = [SOME(true), SOME(true), NONE, SOME(true)]
fun divide (NONE, _) = NONE
  | divide (_, NONE) = NONE
  | divide (SOME(x), SOME(y)) = if y = 0.0 then NONE else SOME(x / y)
fun pair_some (SOME(x), SOME(y)) = SOME(x, y)
  | pair_some _ = NONE
fun divide2 (r, s) =
    case (pair_some (r, s))
      of NONE => NONE
       | (SOME(x, y)) => if y = 0.0 then NONE else SOME(x / y)

Znany nam wbudowany typ list jest rekurencyjny i polimorficzny i ma nastepujaca definicje:

infixr 5 ::
datatype ’a list = nil | :: of (’a * ’a list)

A drzewo można zdefiniować na przykład tak:

datatype ’a tree = EMPTY | NODE of ’a tree * ’a * ’a tree

Zadania

Zadanie 4.1 Napisz wielkie mnóstwo funkcji na drzewach. Możesz napisać odpowiedniki wszystkich znanych Ci funkcji na listach. Możesz znaleźć jakieś funkcje specyficzne dla drzew, na przykład: obchodzace drzewo, zamieniajace prawe poddrzewa z lewymi, etc. Możesz napisać funkcje z drzew w listy i z list w drzewa.

Zadanie 4.2 Zaimplementuj sortowanie drzewowe (treesort).

Wskazówka: W uporzadkowanym drzewie binarnym liczba wystepujaca w weźle jest wieksza od wszystkich w lewym poddrzewie i mniejsza od wszyskich w prawym. Należy zdefiniować dwie główne funkcje. Jedna, która dostaje liste liczb naturalnych, zaś oddaje uporzadkowane drzewo binarne zawierajace elementy tej listy. Druga, która dostaje uporzadkowane drzewo binarne, a oddaje posortowana liste jego elementów. Treesort to złożenie tych funkcji.

Zadanie 4.3 To jest bardzo ładna (i odrobinke nieefektywna) definicja liczb naturalnych:

    datatype nat = ZERO | SUCC of nat

SUCC od successor - nastepnik. Napisz kilka podstawowych funkcji i relacji

(czyli funkcji typu nat->bool) na tych liczbach naturalnych.

Zadanie 4.4 Oto drzewo składni abstrakcyjnej (gramatyka) pewnego jezyka wyrażeń

datatype expression =
    NUM of int
  | TIMES of expression * expression
  | PLUS of expression * expression

Napisz funkcje wyliczajaca wartość wyrażenia. Napisz wersje LL1 powyższej gramatyki. Napisz funkcje przekształcajaca wyrażenie w gramatyce LL1 w wyrażenie w podanej wyżej gramatyce. Napisz parser (używajac gramatyki LL1) i lekser. Wskazówka:

<1st_atom> -> <numeral>
<2nd_atom> -> <1st_atom> <2nd_atom’>
<2nd_atom’> -> epsilon | * <1st_atom> <2nd_atom’>
<3rd_atom> -> <2nd_atom> <3rd_atom’>
<3rd_atom’> -> epsilon | + <2nd_atom> <3rd_atom’>
<expression> -> <3rd_atom>

Zadanie 4.5 Zdefiniuj typ drzew, w których każdy wierzchołek może mieć dowolnie wielu synów. Napisz troche funkcji na tych drzewach.

Rozdział 5

Funkcje wyższych rz edów

Oto definicja funkcji sum:

fun sum [] = 0
  | sum (a::rest) = a + sum rest

Oto definicja funkcji prod:

fun prod [] = 1
  | prod (a::rest) = a * prod rest

Te funkcje sa podejrzanie podobne.

Różnia sie tylko nazwa i tym, że w deficji sum w pierwszym równaniu wystepuje 0 a w drugim +, podczas gdy w definicji prod, w tych samym miejscu w pierwszym równaniu jest 1 a w drugim *. Chciałoby sie napisać funkcje foldr taka, żeby sum = foldr (+, 0) i prod = foldr (∗, 1). Nic prostszego:

fun foldr (f, x) [] = x
  | foldr (f, x) (a::rest) = f(a, foldr (f, x) rest)

i tak powstała pierwsza funkcja wyższego rzedu ;-).

Teraz można napisać:

val sum = foldr (op +, 0) (* op, because + is an infix identifier *)
val prod = foldr (op *, 1)

Zauważmy, że foldr tak jak go zapisaliśmy jest polimorficzny:

val sum_reals = foldr (op +, 0.0)
fun and_pair (b, c) = b andalso c
val and_list = foldr (and_pair, true)
val list_identity = foldr (op ::, nil)
val add_sizes = foldr (fn (s, rest_size) => (size s) + rest_size, 0)

(fn x => E, gdzie E to jakieś wyrażenie (być może zawierajace x), oznacza funkcje która x przyporzadkowuje E, jest to tak zwana anonimowa funkcja zapisana w tak zwanej lambda notacji).

Dla wygody przyjeto, że wyrażenie w rodzaju (f a) b (aplikacja funkcji

(f a) do argumentu b) można zapisywać bez nawiasów: f a b. Przykład:
(foldr (op +, 0)) [2, 44, 5] można zapisać jako: foldr (op +, 0) [2, 44, 5].

Na zakończenie obrazowe podsumowanie działania foldr:

foldr (f, x)[a_1, a_2, a_3, a_4] = f(a_1, f(a_2, f(a_3, f(a_4, x)))) lub w nieco czytelniejszym, infiksowym zapisie: foldr (+, x)[a_1, a_2, a_3, a_4] = a_1 + (a_2 + (a_3 + (a_4 + x)))

Zadania

Zadanie 5.1 Przy pomocy foldr napisz append. Zadanie 5.2 Przy pomocy foldr napisz length.

Zadanie 5.3 Przy pomocy foldr napisz funkcje good_max, która dostawszy

n : int oraz l : int list daje najwieksza liczbe z listy l wieksza od
n, lub n jeśli w l nie ma żadnej liczby wiekszej od n.

Zadanie 5.4 Napisz funkcje map, która dostaje funkcje f i liste l, a daje liste wyników aplikacji funkcji f do kolejnych elementów l. Piszac prosto: map f [a_1, a_2, a_3, a_4] = [f a_1, f a_2, f a_3, f a_4].

Ten sam map zapisz przy pomocy foldr.

Zadanie 5.5 Przy pomocy map napisz funkcje first_cut, która dostaje liste list a daje liste pierwszych elementów tych list. Napisz first_cut nie używajac map. Czy ta wersja nie przypomina Ci definicji map?

Zadanie 5.6 Napisz funkcje filter, która dostaje predykat p : ’a -> bool i liste l : ’a list i daje liste składajaca sie z tych elementów l, które spełniaja p.

Napisz filter korzystajac z foldr. Czy ta definicja jest bardziej czytelna?

Zadanie 5.7 Przy pomocy filter napisz divis_list, która wyławia z listy te liczby które sa podzielne przez zadana liczbe. Prawda, że ładnie?

Zadanie 5.8 Napisz foldl, czyli funkcje która zachowuje sie w ten sposób foldl (f, x)[a_1, a_2, a_3, a_4] = f(f(f(f(x, a_1), a_2), a_3), a_4) lub w infiksowym zapisie:

foldl (+, x)[a_1, a_2, a_3, a_4] = ((((x + a_1) + a_2) + a_3) + a_4) Zadanie 5.9 Używajac miedzy innymi foldl napisz append. Zwykły append. Zadanie 5.10 Używajac miedzy innymi foldl napisz foldr. Korzystaj z doświadczeń z poprzedniego zadania.

Zadanie 5.11 Napisz odpowiedniki foldr, map i filter dla drzew.

Rozdział 6

Funkcje jeszcze wyższych rz edów

Funkcje wyższych rzedów spotyka sie wszedzie. Weźmy choćby aplikacje funkcji do argumentu. Toć to twór wyższego rzedu! Możemy go nazwać:

fun apply (f, a) = f a

i od tej pory zawsze pisać apply (g, b) tam gdzie dotad pisaliśmy g b :-].

Operacja składania dwóch funkcji jest oczywiście funkcja wyższego rzedu. Pewna jej wersja jest nazwana i zdefiniowana w standardowej bazie SML-a:

infix 3 o
fun (f o g) x = f (g x)

i czasem sie przydaje, np.:

val treesort = bst2list o list2bst

Nawet pospolite operacje arytmetyczne można zapisać jako funkcje wyższego rzedu:

fun divide_by x y = y / x;
divide_by 2.0 4.0;
val divide_by_two = divide_by 2.0;
divide_by_two 10000000.0;
divide_by_two 342423423452345.0;
map (divide_by 3.0) [333.0, 12.0, 55.0, 5.0];

Definiujac divide_by na podstawie funkcji / dokonaliśmy pewnych myślowych operacji wyższego rzedu (jakżeby inaczej). Te operacje można zapisać w SML-u:

fun curry f a b = f (a, b)
fun C f x y = f y x

Definicje divide_by można teraz przedstawić tak:

val divide_by = C (curry (op /))

Tak na marginesie: dla dowolnego f : ’a * ’b -> ’c istnieje dokładnie jedna funkcja g : ’a -> ’b -> ’c taka, że dla każdego (x, y) : ’a * ’b

f (x, y) = g x y. Tym g jest curry f. Skad wynika na przykład, że każda

operacje arytmetyczna można przerobić na funkcje wyższego rzedu tylko na dwa sposoby.

Jeszcze jedno — przy pomocy lambda notacji funkcje wyższych rzedów zapisuje sie zagnieżdżajac fn, np.:

fun curry f = fn a => fn b => f (a, b)
val C = fn f => fn x => fn y => f y x

Zadania

Zadanie 6.1 Zdefiniuj funkcje odwrotne do curry (bedzie sie nazywała uncurry) i C (nazwe wymyśl sama (sam)). Napisz funkcje repeat, która dostawszy pare (f, n) daje funkcje równa złożeniu n funkcji f. Czyli:

repeat (f, n) = f^n

Zadanie 6.2 Dla dowolnych dwóch typów ’a i ’b istnieje typ ’a * ’b, ich produkt. W naturalny sposób zwiazane sa z tym typem pewne funkcje — projekcje:

fun pi1 (x, y) = x
fun pi2 (x, y) = y

i funkcja pair typu (’c -> ’a) * (’c -> ’b) -> (’c -> ’a * ’b), która dostawszy funkcje f :’c -> ’a oraz g : ’c -> ’b daje funkcje

h : ’c -> ’a * ’b, o tej własności, że pi1 o h = f i pi2 o h = g.

(Taka funkcja h jest dla każdego (f, g) tylko jedna.)

Napisz funkcje pair.

Zadanie 6.3 Zdefiniuj typ (’a, ’b) direct_sum, który jest suma rozłaczna tyów ’a i ’b. Ten typ bedzie w pewnym sensie dualny do typu ’a * ’b. Zdefiniuj włożenia (injekcje):

inleft : ’a -> (’a, ’b) direct_sum
inright : ’b -> (’a, ’b) direct_sum

i funkcje cases analogiczna (tak naprawde dualna) do funkcji pair dla produktu.

Wskazówka 1: Funkcja cases jest typu

(’a -> ’c) * (’b -> ’c) -> (’a, ’b) direct_sum -> ’c.

Wskazówka 2: Funkcja cases dostawszy funkcje f : ’a -> ’c oraz

g : ’b -> ’c daje funkcje h : (’a, ’b) direct_sum -> ’c , o tej

własności, że h o inleft = f i h o inright = g . (Taka funkcja h jest dla każdego (f, g) tylko jedna.)

Zadanie 6.4 Napisz funkcje

exists : (’a -> bool) -> ’a list -> bool

która bierze predykat p : (’a -> bool) i daje funkcje, która bierze liste

l i oddaje true gdy któryś z elementów listy spełnia predykat p, false w

przeciwnym przypadku. Napisz funkcje

non : (’a -> bool) -> (’a -> bool)

która neguje predykat. Napisz funcje forall, która tym sie różni od

exists, że w odpowiednim miejscu nie któryś, ale wszystkie elementy

listy musza spełniać predykat. Nim zaczniesz pisać forall podobnie jak

exists, pomyśl czy nie możnaby go napisać śmieszniej (nie, nie chodzi mi

o to żeby użyc foldr).

Spoiler: W definicji forall korzystaj wyłacznie z exists, non, i o.

Zadanie 6.5 Zapisz insertionsort lub inny algorytm sortowania (moźe mergesort, jeśli jeszcze go nie napisałeś lub nie napisałaś) w takiej postaci, aby jako pierwszy argument brał funkcje le : ’t * ’t -> bool wyznaczjaca pewien porzadek i sortował liste l : ’t list według tego porzadku. Napisz le_int i le_string. Wypróbuj insertionsort le_int. Napisz też jakiś

le_int_to_int : (int -> int) * (int -> int) -> bool

Wypróbuj insertionsort le_int_to_int.

Rozdział 7

Operacje na tekstach

Operacje na tekstach w SML to w rzeczywistości operacje na listach znaków. Istnieje wbudowana funkcja explode, która dostajac napis, daje w wyniku liste znaków, z których sie on składa. Np. explode "baba" daje

[#"b",#"a",#"b",#"a"]

lub

["b","a","b","a"]

w zależnosci od implementacji.

Zadania

Zadanie 7.1 Używajac funkcji explode zdefiniuj funkcje size, ktora oblicza długość napisu.

Zadanie 7.2 implode to funkcja odwrotna do explode. Napisz funkcje łaczaca dwa napisy.

Zadanie 7.3 Napisz funkcje linesplit : string -> string list, przerabiajaca napis na liste wierszy, z których sie on składa.

Zadanie 7.4 Napisz funkcje wordsplit : string -> string list, przerabiajaca napis na liste słów, z których sie on składa.

Zadanie 7.5 Napisz funkcje

generic_split : string -> string -> string list,

przerabiajaca napis na liste pól, z których sie on składa, przyjmujac drugi argument funkcji za zbiór separatorów pól.

Zadanie 7.6 Napisz funkcje sortujaca leksykograficznie liste słów (można, a nawet należy skorzystać z napisanej wcześniej funkcji sortujacej) Zadanie 7.7 Napisz funkcje generujaca dla danego napisu liste n najcześciej w nim wystepujacych słów, wraz z liczba wystapień każdego z nich.

Zadanie 7.8 Napisz funkcje

strchr : string * char -> int list

taka, że strchr(s,c) daje liste pozycji, na których znak c wystepuje w napisie s.

Zadanie 7.9 Napisz funkcje

strstr : string * string -> int list

taka, że strstr(s,p) daje liste pozycji, na których zaczynaja sie wystapienia napisu p w napisie s. Uwaga:

strstr("ababababa", "babab") = [2,4].

Leniwość

Rozdział 9

Moduły: struktury i sygnatury

Każdy duży a dobry program w naturalny sposób dzieli sie na szereg odrebnych, w sporym stopniu niezależnych od siebie podprogramów. (Czesto podział ten nastepuje już w fazach projektowania i specyfikowania.) Każdy taki podprogram składa sie z pewnej liczby typów, wartości, komentarzy, intuicji, itp. i wszystkie jego składniki sa mocno ze soba logicznie powiazane. Jeśli zamanifestujemy istnienie takiego podprogramu deklarujac odpowiedni SML-owy moduł, to zarówno dla nas, jak i dla kompilatora stanie sie lepiej widoczne, co dokładnie wchodzi w skład podprogramu, jakie sa zależnosci miedzy jego komponentami, itd. Jeśli dodatkowo przyjrzymy sie krytycznie powstałemu (lub projektowanemu) modułowi i ocenimy co w nim jest istotne z punktu widzenia reszty naszego programu, a co jest jedynie szczegółem implementacyjnym, to bedziemy mogli opatrzyć go wzglednie krótka, a mimo to dobra, sygnatura.

Przykład, który wiele wyjaśnia:

signature TOTAL_PREORDER =
sig

    type elem
    val leq : elem * elem -> bool
    (* and leq should be a total preorder *)

end;

signature PRIORITY_QUEUE =
sig

    structure Item : TOTAL_PREORDER

    type queue

    val empty : queue
    val insert : Item.elem * queue -> queue
    val join : queue * queue -> queue

    datatype min = MIN of Item.elem * queue | EMPTY
    val get_min : queue -> min

    (* By seq(q) we denote the sequence of Item.elem elements
       produced by iterating get_min on queue q.
       By num(t, q) we denote the number of occurrences of t in seq(q).

       For all q, p - queues constructed using the above operations,
       and for all t, u - elements of type Item.elem:
       1. seq(q) should be finite and non-decreasing,
       2. num(t, empty) should be 0,
       3. num(t, insert(u, q)) should be num(t, q), if u <> t,
          or num(t, q) + 1, if u = t,
       4. num(t, join(q, p)) should be num(t, q) + num(t, p). *)

end;

signature INTEGER_LEQ =
sig

    include TOTAL_PREORDER
    sharing type elem = int
    (* and leq should be the natural less-equal on int *)

end;

structure IntegerLeq : INTEGER_LEQ =
struct

    type elem = int
    fun leq (k, l) = ((k : int) <= l)

end;

signature INT_PRIORITY_QUEUE =
sig

    include PRIORITY_QUEUE
    sharing type Item.elem = int
    (* and Item.leq should be the natural less-equal on int *)

end;

structure IntPQasList : INT_PRIORITY_QUEUE =
struct

    structure Item = IntegerLeq (* type sharing and requirement *)
                                (* about Item.leq now satisfied *)

    type queue = Item.elem list (* queues will be unordered lists *)

    val empty = nil
    val insert = op ::
    val join = op @

    datatype min = MIN of Item.elem * queue | EMPTY
    fun get_min nil = EMPTY
      | get_min (k::tail) =
        case get_min tail
          of EMPTY => MIN(k, nil)
           | MIN(n, rest) => if Item.leq (k, n) then MIN(k, n::rest)
                             else MIN(n, k::rest)

end;

Nazwijmy naszkicowany powyżej program IntP QasList. Jest on poprawny tzn. dobrze implementuje kolejke priorytetowa liczb całkowitych. W szczególności IntPQasList pasuje do INT_PRIORITY_QUEUE, czyli zawiera wszystkie wymagane typy, wartości, itp., jak również spełnia wszystkie wymagania opisane w komentarzach (sprawdź).

Definicja (przez przykład) dobroci: Mówimy, ze sygnatura INTEGER_LEQ jest dobra dla struktury IntegerLeq w programie IntP QasList, gdyż:

1. IntegerLeq pasuje do INTEGER_LEQ

2. jesli napiszemy dowolna strukture A pasujaca do INTEGER_LEQ, to program IntP QasList z IntegerLeq podmienionym na A pozostanie poprawny

Dobra sygnatura pozwala nam i kompilatorowi rozumować o module, jako całości, wyłacznie na podstawie jego sygnatury, bez zagladania do kodu. Wspomaga to abstrakcyjne myślenie o programie, bardzo ułatwia pielegnacje i modyfikacje, pozwala na rozłaczna kompilacje modułów, etc.

Gdyby w programie IntP QasList zamiast INTEGER_LEQ zdefiniować nastepujaca sygnature:

signature INTEGER_LEQ’ =
sig

    include TOTAL_PREORDER
    sharing type elem = int
    (* and leq should additionally be antisymmetric *)

end;

to nie byłaby ona dobra sygnatura dla IntegerLeq. Oto dowód:

structure A : INTEGER_LEQ’ =
struct

    type elem = int
    fun leq (k, l) = ((k : int) >= l) (* comment satisfied *)

end;

Widać, że A pasuje do INTEGER_LEQ’. Ale program IntP QasList z INTEGER_LEQ’ zamiast INTEGER_LEQ, gdy podmienimy w nim IntegerLeq na A, przestaje być poprawny, gdyż porzadek na Item.elem w module IntPQasList jest odwrotny niż wymagany w jego specyfikacji.

Pisz zawsze dobre sygnatury!

Zadania

Zadanie 9.1 Napisz moduły COMPLEX_LEQ (liczby zespolone uporzadkowane według ich modułów), COMPLEX_PRIORITY_QUEUE, ComplexLeq, ComplexPQasList. Sprawdź poprawność i przetestuj.

Zadanie 9.2 Napisz program IntP QasOrderedList. Upewnij sie, że moduł

IntPQasOrderedList pasuje do sygnatury INT_PRIORITY_QUEUE. Pomyśl

o złożoności czasowej IntP QasList oraz IntP QasOrderedList. Zadanie 9.3 Napisz program sortujacy liczby całkowite przy użyciu kolejki priorytetowej. Wykorzystaj IntP QasList jako jego podprogram. Czy

INT_PRIORITY_QUEUE jest dobra sygnatura dla IntPQasList w tym pro-

gramie?

Zadanie 9.4 Zmodyfikuj program z poprzedniego zadania tak, aby korzystał z IntP QasOrderedList. Jaki algorytm sortowania wynika z użycia insert a jaki z użycia join przy tworzeniu kolejki?

Rozdział 10

Moduły: funktory

Przy pomocy funktora można wyraźnie opisać, jakie moduły sa potrzebne by zaimplementować dany moduł. Gdy program jest napisany w całości przy pomocy funktorów, w naturalny sposób przybiera on hierarchiczna postać. Jest jeden moduł, który rzadzi i którego sygnatura jest opisem naszych oczekiwań co do programu, sa moduły spełniajace specyfikacje argumentów tego rzadzacego funktora, sa moduły spełniajace specyfikacje argumentów tych modułów, itd. Dodatkowo, jeśli funktor ma dobra sygnature i nie wymienia sie w nim nazw żadnych zewnetrznych modułów (element dobrego stylu), to można o nim rozumować lub zmieniać jego implementacje, abstrahujac od własności czy istnienia jakichkolwiek innych modułów i nie psuje to poprawności programu.

Przykład funktora:

... (* <- TREE and Tree are defined here *)

functor PQasHeap
    ( structure TotalPreorder : TOTAL_PREORDER
      structure Tree : TREE ) :
sig

    include PRIORITY_QUEUE
    sharing Item = TotalPreorder

    (* queues should be constructed as trees *)
    (* with value of every node less or equal *)
    (* to values of all the descendant nodes *)

end =
struct

    structure Item = TotalPreorder (* sharing satisfied *)

    type queue = Item.elem Tree.tree
    ... (* still some work ;-) *)

end;

structure IntPQasHeap = PQasHeap (structure TotalPreorder = IntegerLeq
                                  structure Tree = Tree);

Zadania

Zadanie 10.1 Napisz program P QasHeap według szkicu podanego powyżej.

Sprawdź, czy sygnatury sa dobre. Przetestuj w przypadku liczb całkowitych i zespolonych.

Zadanie 10.2 Zmień TREE i Tree tak, by Tree był funktorem, o argumencie Item, a tree nie było polimorficznym typem danych, tylko typem zależnym od argumentu funktora:

datatype tree = EMPTY | NODE of tree * Item.item * tree.

Użyj tego TREE i Tree do zrobienia innej wersji programu P QasHeap. To jest niezbyt ładne bez funktorów wyższego rzedu, ale wykonalne.

Zadanie 10.3 Zmodyfikuj P QasHeap tak, by drzewa, jakimi sa kolejki, były zawsze utrzymywane w postaci wyważonej. Postaraj sie w jak najwiekszym fragmencie programu abstrahować od sposobu w jaki zapewniane jest wyważenie. Zamanifestuj te abstrakcje odpowiednio deklarujac moduły.

Zadanie 10.4 Napisz funktor implementujacy sortowanie przy pomocy kolejek priorytetowych. Pomyśl o złożoności czasowej sortowania przy użyciu P QasHeap.

To już koniec :-[

Dodatek A

Jak zaliczyć laboratorium

Aby zaliczyć laboratorium SML należy napisać program. Powinien sie on składać z kilku do kilkunastu małych modułów; dobrze wyspecyfikowanych, połaczonych w sensowna całość jasnymi, wyraźnie opisanymi zależnościami.

Temat jest dowolny. Służymy rada i pomoca przy jego wyborze. Oto przykładowe (troche nudne i wykorzystywane już zbyt wiele razy) zadanie zaliczeniowe:

Napisać Kalkulator, czyli program, który wśród swoich modułów ma moduł pasujacy do sygnatury CALCULATOR, zdefiniowanej nastepujaco:

signature CALCULATOR =
sig

    val eval : string -> string

    (* if s represents an arithmetic expression *)
    (* (which means that s is a string of digits, *)
    (* ~ (negation), +, -, * and parenthesis, *)
    (* satisfying some well known properties), *)
    (* then (eval s) should be equal to the string *)
    (* representing the value of s, or the string *)
    (* "division by zero" if appropriate, *)
    (* if s does not represent a valid arithmetic *)
    (* expression, (eval s) should be equal to the *)
    (* string "syntax error" *)

end;

Wszelkie drobne niejasności tej specyfikacji należy interpretować na swoja korzyść. W razie problemów zwrócić sie o pomoc.

Wydaje sie, że wśród modułów składajacych sie na Kalkulator powinny być moduły Lexer, Parser, Evaluator i PrettyPrinter. Byłoby ładnie i chyba wygodnie, gdyby Parser posługiwał sie drzewem składni reprezentujacym gramatyke LL(1) wyrażeń, zaś Evaluator drzewem reprezentujacym prostsza gramatyke:

datatype expression =
    NUM of int
  | NEGATION of expression
  | TIMES of expression * expression
  | PLUS of expression * expression
  | ...

To zadanie nie powninno być zbyt trudne. To co można tu zrobić lepiej lub gorzej to podział programu na moduły, zdefiniowanie zależności miedzy modułami przy pomocy funktorów, opisanie własności modułów sygnaturami.

Powodzenia!

Dodatek B

Rozwi azania niektórych zadań

Wszystkie te rzeczy przeszły przez kompilator (w wypadku programów w SML to bardzo dobry omen), ale nie były zbyt intensywnie testowane. Dzieki za wskazywanie błedów!

(* Exercise 2.1 *)
exception Empty
fun tl [] = raise Empty
  | tl (head::tail) = tail
(* Exercise 2.2 *)
fun sum l = if null l then 0 else hd l + sum (tl l)
(* or *)
fun sum [] = 0
  | sum (i::rest) = i + sum rest
(* Exercise 2.3 *)
fun length l = if null l then 0 else 1 + length (tl l)
(* or *)
fun length [] = 0
  | length (head::tail) = 1 + length tail
(* Exercise 2.4 *)
fun last l = if null l then hd l else last (tl l)

(* or *)

fun last [] = raise Empty
  | last [elem] = elem
  | last (head::tail) = last tail

(* Exercise 2.5 *)

fun append ([], l) = l
  | append (head::tail, l) = head :: (append (tail, l))

(* Exercise 2.6 *)

fun take (l, 0) = nil
  | take (head::tail, n) = head :: take (tail, n-1)
  | take (nil, n) = nil

fun drop (l, 0) = l
  | drop (head::tail, n) = drop (tail, n-1)
  | drop (nil, n) = nil

(* Check, that the required equality *)
(* holds even if n < 0. This is an ugly hack, *)
(* I know, but I hate exceptions even more:P. *)

(* Exercise 2.7 *)

fun nth (l, n) = hd (drop (l, n-1))

(* And this version is slower. Why? *)

fun nth2 (l, n) = last (take (l, n))

(* Exercise 2.8 *)

fun flatten [] = []
  | flatten (l::rest) = append (l, flatten rest)

(* Exercise 2.9 *)

(* @ is the built in, infix version of append *)

fun quicksort [] = [] : int list
  | quicksort [x] = [x]
  | quicksort (a::rest) =
    let
        fun split (b, []) = ([], [])
          | split (b, x::l) = let
                                  val (left, right) = split (b, l)
                              in

                                  if x < b then (x::left, right)
                                  else (left,x::right)
                              end
    in
        let
            val (left, right) = split (a, rest)
        in
            quicksort left @ (a::quicksort right)
        end
    end

(* Exercise 3.1 *)

fun length l = let
                   fun len_acc (len, []) = len
                     | len_acc (len, head::tail) = len_acc (len + 1, tail)
               in
                   len_acc (0, l)
               end

(* Exercise 3.2 *)

fun decimal l =
  let fun diggy (acc, []) = acc
        | diggy (acc, digit::rest) = diggy (10*acc + digit, rest)
  in
      diggy(0, l)
  end

(* Definitions of decimal without an accumulator are absolutely hopeless *)

(* Exercise 3.3 *)

(* reverse2 is O(n^2) *)

fun reverse2 [] = []
  | reverse2 (head::tail) = (reverse2 tail) @ [head]

(* Thanks to the use of accumulator reverse is O(n) *)
(* and by chance there is tail recursion here *)

fun reverse l = let
                    fun rev_acc (rev, []) = rev
                      | rev_acc (rev, head::tail) = rev_acc (head::rev, tail)
                in
                    rev_acc ([], l)
                end

(* Exercise 3.4 *)

(* This works. Notice reverse. *)
(* A similar thing could be done for append. *)
(* Not worth the effort, is it? *)

exception Empty

fun take (l, n) =
    let
        fun take_acc (acc, (l, 0)) = reverse acc
          | take_acc (acc, (hd::tl, n)) = take_acc (hd :: acc, (tl, n-1))
          | take_acc (acc, (nil, n)) = raise Empty
    in
        take_acc ([], (l, n))
    end

(* Exercise 3.5 *)

fun interval (from, to) =
    let
        fun inter_acc (l, (n, m)) = if n > m then l
                                    else inter_acc (m::l, (n, m-1))
    in
        inter_acc ([], (from, to))
    end

(* Exercise 3.6 *)

val minus_infinity = ~10000000000000000000000000000000.0

fun real_max [] = minus_infinity
  | real_max (head::tail) =
    let fun maxx (r, []) = r
          | maxx (r, x::tl) = if r > x then maxx (r, tl)
                              else maxx (x, tl)
    in
        maxx (head, tail)
    end

fun real_max [] = minus_infinity
  | real_max (x::tail) = let
                        val y = real_max tail
                    in
                        if y > x then y
                        else x
                    end

(* Exercise 5.1 *)
fun foldr (f, x) [] = x
  | foldr (f, x) (a::rest) = f(a, foldr (f, x) rest)
fun append (l1, l2) = foldr (op ::, l2) l1
(* Exercise 5.2 *)
val length = foldr (fn (elem, rest_length) => 1 + rest_length, 0)
(* Exercise 5.3 *)
fun good_max n = foldr (max, n)
(* Exercise 5.4 *)
fun map f nil = nil
  | map f (head::tail) = (f head)::(map f tail)
fun map f = foldr (fn (a, result) => (f a):: result, nil)
(* Exercise 5.5 *)
val first_cut = map hd
fun first_cut nil = nil
  | first_cut (head::tail) = (hd head)::(first_cut tail)
(* Exercise 5.6 *)
fun filter p nil = nil
  | filter p (a::rest) = if p a then a::(filter p rest)
                         else (filter p rest)
fun filter p = foldr (fn (a, filtered_tail) =>
                      if p a then a::filtered_tail
                      else filtered_tail, nil)
(* No, this didn’t turn out more readable (at least for me). *)
(* Exercise 5.7 *)
fun divisible_by k l = l mod k = 0
fun divis_list n = filter (divisible_by n)
(* Yeees. *)
(* Exercise 5.8 *)
fun foldl (f, x) nil = x
  | foldl (f, x) (head::tail) = foldl (f, (f (x, head))) tail

(* Exercise 5.9 *)

fun reversed_cons (tail, head) = head :: tail
fun append (l1, l2) = foldl (reversed_cons, l2) (rev l1)

(* Exercise 6.1 *)

fun uncurry f = fn (a, b) => f a b

fun id x = x
fun repeat (f, 0) = id
  | repeat (f, n) = f o (repeat (f, n - 1))

(* Exercise 6.2 *)

fun pair (f, g) x = (f x, g x)

(* Exercise 6.3 *)

datatype (’a, ’b) direct_sum = INL of ’a | INR of ’b
val inleft = INL
val inright = INR
fun cases (f, g) (INL(x)) = f x
  | cases (f, g) (INR(y)) = g y

(* Exercise 6.4 *)

fun exists p [] = false
  | exists p (a::rest) = if p a then true
                         else exists p rest
fun non p = not o p
val forall = non o exists o non

Dodatek C

Jak pracować z kompilatorem SML-a

• Interaktywnie:

1. uruchomić SML (mosml, new-sml, sml-0.93, sml) 2. po zachecie, która na ogół jest myślnik, napisać program 3. zakończyć pisanie średnikiem i nacisnać RET • Wsadowo:

1. napisać program jakimś edytorem 2. zapisać go w pliku smth.sml 3. uruchomić SML i napisac use "smth.sml";

• Przy pomocy EMACS-a i sml-mode:

1. uruchomić EMACS-a 2. C-x C-f smth.sml RET (ważne, żeby końcówka nazwy było .sml) 3. napisać błedny program (Tab automagicznie formatuje) 4. C-c C-l i bedzie wiadomo co dalej 5. Zrobić C-c ‘ (ten ostatni znaczek to back-quote, wystepuje na klawiaturze na ogół pod tylda), wtedy on ustawi kursor tam gdzie w pliku jest bład 6. Korzystać z innych dobrodziejstw sml-mode, np. sml-mode-info, czyli podrecznika do sml-mode